unittest
{
	assert( [ "4 4", "1 2", "2 4", "1 3", "3 4" ].parse.expand.solve == "YES" );
	assert( [ "5 5", "1 2", "3 5", "5 1", "3 4", "2 3" ].parse.expand.solve == "NO" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nm = input.front.split.to!( long[] );
	auto abs = new long[ 2 ][ nm[ 1 ] ];
	foreach( ref ab; abs )
	{
		input.popFront;
		ab = input.front.split.to!( long[] );
		ab[] -= 1;
	}
	return tuple( nm[ 0 ], abs );
}

auto solve( long n, long[ 2 ][] abs )
{
	auto cs = new long[ n ];
	foreach( ab; abs )
	{
		cs[ ab[ 0 ] ]++;
		cs[ ab[ 1 ] ]++;
	}
	return cs.count!( ( x ) => x % 2 == 1 ) ? "NO" : "YES";
}
